

一种计算LD₅₀的BASIC语言程序

浙江医科大学

章元沛 方瑞英

杭州自动化研究所

洪一池

LD₅₀(半数致死量)测定为药学研究中的经常性工作。计算LD₅₀值以加权直线回归法为最精确。惜乎其步骤繁复,因而有时不得不采用种种简算法。此种计算烦琐问题目前已可用电子计算机来解决。我们根据SHARP PC-1500袖珍机的性能特点,用BASIC语言编制了按加权直线回归法原理^[1,2],计算LD₅₀值的程序,颇为实用,兹作介绍。

一、计算公式及有关代号

N: 组内动物数

X: 各组剂量的对数

Y: 由各组死亡百分率转换而来的概率单位数

$$W(\text{权重}) = W_c(\text{权重系数}) \times N$$

$$\bar{X} = \frac{\sum WX}{\sum W} \quad \bar{Y} = \frac{\sum WY}{\sum W}$$

$$\begin{aligned} \sum W(X - \bar{X})(Y - \bar{Y}) \\ = \sum WXY - \frac{\sum WX \cdot \sum WY}{\sum W} \end{aligned}$$

$$\sum W(X - \bar{X})^2 = \sum WX^2 - \frac{(\sum WX)^2}{\sum W}$$

$$b = \frac{\sum W(X - \bar{X})(Y - \bar{Y})}{\sum W(X - \bar{X})^2}$$

$$Y_K = \bar{Y} + b(X - \bar{X})$$

$$\log LD_{50} = m = \frac{5 - (\bar{Y} - b\bar{X})}{b}$$

$$LD_{50} = \log^{-1} m$$

$$Sm = \frac{1}{b} \sqrt{\frac{(m - \bar{X})^2}{\sum W(X - \bar{X})^2} + \frac{1}{\sum W}}$$

LD₅₀的95%平均可信限 = LD₅₀

$$\pm \frac{\log^{-1}(m + 1.96Sm) - \log^{-1}(m - 1.96Sm)}{2}^{[3]}$$

LD₅₀的平均可信限率

$$= \frac{LD_{50} \text{的高限} - LD_{50} \text{的低限}}{2 \times LD_{50}}$$

二、纳入程序的数据检索表

1. 1%至99%反应率的概率单位和权重系数表^[4]

2. 0%或100%反应率的概率单位近似值和权重表^[6]

三、程序框图(图1)




图1 计算LD₅₀的程序框图

四、程序

1: DATA 53,29,20,17,13,12,10,9,9,
9,7,6,6,5,5,5,5,4,4,4,3.5,4,3.5,
3,4,4,3.5,3,3,3

```

2: DATA 266, 2.4, 290, 2, 310, 1.5,
   325, 1.4, 339, 1.3, 352, 1.2, 364, 1.2,
   376, 0
3: DATA 664, 23, 14, 9, 8, 6, 5, 4, 4, 3,
   3, 3, 3, 2, 2, 2, 2, 1.5, 1.5, 1.5, 1.5,
   1.5, 1.5, 1, 1, 1, 1, 1
4: DATA 774, .9, 783, .7, 790, .5, 795,
   .4, 799, .4, 803, .4, 807, .3, 810, 0
5: DATA 267, 28, 17, 13, 11, 9, 7, 7, 7,
   6, 5, 6, 4, 5, 4, 5, 4, 3, 4, 4, 3, 4, 3, 3, 4,
   3, 3, 3, 3, 2, 3, 3
6: DATA 459, 2, 3, 3, 2, 3, 3, 2, 3, 2, 3,
   2, 3, 2, 3, 2, 3, 3, 2, 3, 2, 3, 2, 3, 2,
   3, 3, 2, 3, 3, 2
7: DATA 544, 3, 3, 2, 3, 3, 3, 3, 3, 4, 3,
   3, 4, 3, 4, 4, 3, 4, 5, 4, 5, 5, 5, 5, 6, 7, 7,
   7, 9, 11, 13, 17, 28
8: DATA 71, 50, 38, 35, 31, 27, 24, 25,
   21, 21, 17, 19, 16, 17, 13, 17, 13, 12,
   11, 12, 10, 12, 8, 9, 11
9: DATA 548, 7, 8, 7, 6, 5, 6, 6, 6, 3, 6,
   4, 3, 3, 4, 2, 3, 2, 2, 2, 1, 1, 1, 0, 1
10: GRAPH:CSIZE 1:COLOR 3:GLCURS
   OR (40, 0):LPRINT "LD 50":
   COLOR 0:GLCURS OR (0, -20):SO
   RGN
15: INPUT "No. of Dose Groups:";
   G:E=G-1:DIM X(E), Y(E)
20: L=0:H=0:X=0:Y=0:W=0:Z
   =0:R1=0:FOR I=0 TO E:INPUT
   "Dose:"; X(I)
25: GLCURSOR (0, -10*I):LPRINT
   I+1; ".D="; X(I):X(I)=LOG
   X(I):IF X(I)<X(L) THEN LET
   L=I
30: INPUT "No. of Animals:"; N1:GLC
   URSOR (92, -10*I):LRRINT
   "N="; N1:IF X(I)>X(H) THEN
   LET H=I
35: INPUT "No. of Deaths:"; M1:GLC
   URSOR (150, -10*I):LPRINT
   "Nd="; M1:PC=INT ((M1/N1
   +.005)*100)
40: P1=PC:IF (PC=0) OR (PC=100)
   THEN GOTO 70
45: RESTORE 5:IF P1>66 THEN LET
   P1=PC-66:RESTORE 7
50: IF P1>33 THEN LET P1=PC
   -33:RESTORE 6
55: O=P1:GOSUB 151:Y(I)=A/100:
   RESTORE 8:IF PC>50 THEN LET
   PC=100-PC
60: IF PC>25 THEN LET PC=PC
   -25:RESTORE 9
65: O=PC:GOSUB 151:W1=A*
   N1/1000:GOTO 80
70: O=N1:N=1:IF N1>30 THEN LET
   N=2:O=INT (N1/10)
75: RESTORE N:GOSUB 150+N:W1
   =A/100:RESTORE 2+N:GOSUB
   150:Y(I)=A*(PC/50-1)/100
   -PC/10+10
80: X=X+W1*X(I):Y=Y+W1*
   Y(I):W=W+W1:Z=Z+W1*X(I)
   *Y(I):R1=R1+W1*X(I)^2:
   NEXT I:B$="X+"
85: P=X/W:Q=Y/W:B=(Z-X*
   Q)/(R1-X*P):M=(5-Q)/B
   +P:S=SQRT((M-P)^2/(R1-X
   *P)+1/W)/B
90: D=10^M:SC=10^(M-1.96*S)
   *(10^(3.92*S)-1)/2:TEXT:
   CSIZE 1:LF 2:R=Q-B*P:IF
   R<0 THEN LET B$="X"
95: GOSUB 154:R2=R0:R=B:GOSUB
   154:LPRINT "Y="; R0; B$; R2:

```

```

COLOR 3
100: R=D:GOSUB 154:R2=R0:R=SC:
    GOSUB 154:LPRINT "Conf. Lim.
    LD 50="; R2; "+/-" R0
105: COLOR 0:R=SC/D:GOSUB 154:
    LPRINT "Rate of C. L.=",
    RO:LPR INT
110: GRAPH:CSIZE 1:LPRINT "Y":
    LINE (8,0)-(10,10)-(10,-190):
    SORGN:LINE (0,0)-(190,0)
    -(180,2):GLCURSOR (195,0)
115: LPRINT "X":FOR K=0 TO 6
    STEP 2:GLCURSOR (-15,30
    *K):LPRINT K+2
120: NEXT K:FOR K=0 TO 2:T
    =180-60*K:LINE (-2,T)
    -(2,T),0,3
125: NEXT K:GLCURSOR (-10,
    -10):LPRINT INT (X(L)*100
    +0.5)/100:GLCURSOR (175,
    -10)
130: R=X(H):GOSUB 153:LPR INT
    R0:LINE (190,-2)-(190,2),0,
    3:U=190/(X(H)-X(L):FOR
    I=0 TO E
135: V=(X(I)-X(L))*U:LINE (V-2,
    30*(Y(I)-2)-2)-(V+2,30
    *(Y(I)-2)+2),0,3,B
140: NEXT I:LINE (0,30*(Q-2+B
    *(X(L)-P)))-((190,30*(Q-2
    +B*(X(H)-P))),0,2:GLCURS
    OR (-15,90):LPR INT 5
145: F=U*((5-Q)/B+P-X(L)):
    LINE (0,90)-(F,90),5,2:LINE
    (F,90)-(F,0),5,2:GLCURSOR
    (F-22,-10)
150: R=M:GOSUB 155:LPRINT RO:
    GLCURSOR (0,-80):TEXT:END

```

```

151: A=O:FOR I1=1 TO O:READ
    C:A=A+C:NEXT I1:RETURN
152: FOR I2=3 TO O:READ C1,C2:
    A=C1+C2*(N(I)-10*O):NEXT
    I2:RETURN
153: R0=INT (R*100+0.5)/100:
    RETURN
154: R0=INT (R*1000+0.5)/1000:
    RETURN
155: R0=INT (R*10000+0.5)/10000:
    RETURN

```

五、计算实例(图2)

D指剂量, N指组内动物数, Nd指组内死亡动物数, Conf. Lim. 与 Rate of C. L. 分别指LD₅₀的平均可信限与平均可信限率。

六、说明与讨论


1. 本程序中1~9行为《0%或100%反应率的概率单位近似值和权重表》^[1]与《1%至99%反应率的概率单位和权重系数表》^[4]中的数据; 10~105行与151~152行为计算LD₅₀与打印原始数据与计算结果的指令; 110~150行为作图指令; 153~155为按四舍五入法去小数点后尾数指令。如果手边所有计算机只有荧光屏显示结果的设备, 而无打印与绘图附件, 将本程序作些小修改后亦可适用。如: (1) 在10~105行的一段中, 将GRAPH, CSIZE, COLOR等有关语句删去, 把LPRINT改为PRINT。(2) 把110~150行一段全部删去。

2. 由于SHARP PC-1500袖珍机内存不大, 设计程序时应考虑到尽量少占内存, 所以我们对两份数据表作了重新编排。又因该机所带打印机的幅宽有限(有效作图区宽4.5cm), 作图采用了浮动坐标值, 能自动将X的最小值定在极左端, 最大值定在极右端, 不论剂量范围如何, 均能使图象保持最大幅宽。利用该机的多色打印功能, 以区别原始数据与计算结果, 俾能一目了然。

LD 50

1. D= 10	N= 10	Nd= 1
2. D= 12.5	N= 10	Nd= 0
3. D= 15.6	N= 10	Nd= 9

Y= 13.26X-9.424
 Conf.Lim.LD50= 12.239+/- 1.163
 Rate of C.L.= 0.893



LD 50

1. D= 12	N= 5	Nd= 5
2. D= 9	N= 6	Nd= 0
3. D= 7	N= 18	Nd= 11
4. D= 6	N= 34	Nd= 17
5. D= 5	N= 38	Nd= 12
6. D= 4	N= 12	Nd= 2
7. D= 3	N= 8	Nd= 0

Y= 5.083X+ 0.384
 Conf.Lim.LD50= 0.857+/- 0.576
 Rate of C.L.= 0.895




图2 自动打印出来的二个计算LD₅₀结果实例

3. 经在科研实践中试用,体会本程序有下述优点:(1)只需输入各组剂量,组内动物数与死亡数三项原始数据即可。将剂量换算成对数、从死亡率查相应概率单位与权重系数等步骤均能自动进行。(2)实验条件无限制,各组剂量间不成比例,动物数不等,死亡率大至100%,小至0%,亦无妨。(3)可自动打印出回归线方程式、LD₅₀及其平均可信限、平均可信限率三项结果并绘制回归线

图。(4)全过程在2分钟左右的时间内完成,结果精确。

参 考 文 献

- [1] 上海第一医学院卫生统计学教研室编,医学统计方法,209页,上海科技出版社,1979
- [2] 周海钧等编,生物检定统计方法,182页,人民卫生出版社,1983
- [3] 张昌绍、张毅主编,药理学第一卷(总论),323页,人民卫生出版社,1956
- [4] 同上,432页
- [5] 同上,433页